AD-A038 940 WISCONSIN UNIV MADISON MATHEMATICS RESEARCH CENTER F/G 12/1

COMPUTATIONAL ASPECTS OF OPTIMAL RECOVERY.(U)

DEC 76 C DE BOOR DAAG29-75-C-0024

UNCLASSIFIED MRC-TSR-1706 NL

1 OF 1

AD A038940

END

DATE FILMED 5-77

MRC Technical Summary Report #1706

COMPUTATIONAL ASPECTS OF OPTIMAL RECOVERY

Carl de Boor

See 1473

Mathematics Research Center
University of Wisconsin–Madison
610 Walnut Street
Madison, Wisconsin 53706

December 1976

(Received December 3, 1976)

Approved for public release
Distribution unlimited

Sponsored by

U.S. Army Research Office
P.O. Box 12211
Research Triangle Park
North Carolina 27709

UNIVERSITY OF WISCONSIN - MADISON
MATHEMATICS RESEARCH CENTER

COMPUTATIONAL ASPECTS OF OPTIMAL RECOVERY

Carl de Boor

Technical Summary Report #1706
December 1976

ABSTRACT

Recently, Micchelli, Rivlin and Winograd [15] described a scheme for interpolating by splines at given data points which, in a certain reasonable sense, is optimal among all schemes which attempt to recover a function from its values at those data points. This paper offers a Fortran program for the calculation of that optimal interpolant. A short derivation of that recovery scheme is given first, in order to make the paper selfcontained and also to provide an alternative to the original derivation in [15]. For the latter reason, a derivation of the related envelope construction of Gaffney and Powell [8] is also given. From a computational point of view, these schemes are special cases of the following computational problem: to construct an extension with prescribed norm of a linear functional on some finite dimensional linear subspace to all of $\mathbb{L}_1[a,b]$.

AMS (MOS) Subject Classifications: 41A15, 41A05, 41A65, 65D05, 65K05

Key Words: Optimal Recovery, Perfect Splines, Spline Interpolation, Extension of Linear Functionals, Best Approximation of Linear Functionals.

Work Unit Number 6 (Spline Functions and Approximation Theory)

Sponsored by the United States Army under Contract No. DAAG29-75-C-0024.

COMPUTATIONAL ASPECTS OF OPTIMAL RECOVERY

Carl de Boor

Mathematics Research Center, U. Wisconsin-Madison

610 Walnut St., Madison, WI 53706 USA

1. INTRODUCTION

This paper offers a Fortran program for the calculation of the optimal recovery scheme of Micchelli, Rivlin and Winograd [15]. A short derivation of that recovery scheme is given first, in order to make the paper selfcontained and also to provide an alternative to the original derivation in [15]. For the latter reason, a derivation of the related envelope construction of Gaffney and Powell [8] is also given. From a computational point of view, these schemes are special cases of the following computational problem: to construct an extension with prescribed norm of a linear functional on some finite dimensional linear subspace to all of $\mathbb{L}_1[a,b]$.

2. THE OPTIMAL RECOVERY SCHEME OF MICCHELLI, RIVLIN AND WINOGRAD

This scheme concerns the recovery of functions from partial information about them. Let $n > k$ and let $\tau := (\tau_i)_1^n$ be nondecreasing, in some interval $[a,b]$, with $\tau_i < \tau_{i+k}$, all i. For $f \in \mathbb{L}_\infty^{(k)} := \{g \in C^{(k-1)}[a,b] : g^{(k-1)}$ abs. cont., $g^{(k)} \in \mathbb{L}_\infty\}$, denote by $f|_\tau$ the restriction of f to the data point sequence τ, i.e., $f|_\tau := (f_i)_1^n$ with

$$f_i := f^{(j)}(\tau_i),\ j := j(i) := \max\{m : \tau_{i-m} = \tau_i\} .$$

Sponsored by the United States Army under Contract No. DAAG29-75-C-0024.

We call a map $S : \mathbb{L}_\infty^{(k)} \to \mathbb{L}_\infty$ a <u>recovery scheme</u> (<u>on</u> $\mathbb{L}_\infty^{(k)}$) <u>with respect to</u> τ provided Sf depends only on $f|_\tau$, i.e., provided $f|_\tau = g|_\tau$ implies that $Sf = Sg$. The (possibly infinite) constant

$$\text{const}_S := \sup\{\|f - Sf\| / \|f^{(k)}\| : f \in \mathbb{L}^{(k)}\}$$

measures the extent to which such a recovery scheme S may fail to recover some f since it provides the sharp error bound

$$(1)\quad \|f - Sf\| \le \text{const}_S \|f^{(k)}\|, \quad \text{all } f \in \mathbb{L}_\infty^{(k)} .$$

Here and below,

$$\|g\| := \text{ess.sup}\{|g(t)| : a \le t \le b\} .$$

A recovery scheme S is <u>optimal</u> if its const_S is as small as possible, i.e., if the worst possible error is as small as possible, as measured by (1). We write

$$\text{const}_\tau := \inf_S \text{const}_S$$

for the best possible constant. Here is a quick lower bound for that constant. We have

$$\begin{aligned}
\text{const}_S &= \sup_f \sup_{g|_\tau = f|_\tau} \|g - Sf\| / \|g^{(k)}\| \\
&\ge \sup_{g|_\tau = 0} \|g - S0\| / \|g^{(k)}\| \\
&= \sup_{g|_\tau = 0} \max\{\|g - S0\|, \|-g - S0\|\} / \|g^{(k)}\| \\
&\ge \sup_{g|_\tau = 0} \|g\| / \|g^{(k)}\| ,
\end{aligned}$$

for every recovery scheme S, hence

$$(2)\quad \text{const}_\tau \ge c(\tau) := \sup_{g|_\tau = 0} \|g\| / \|g^{(k)}\| .$$

But, actually, equality holds here. For the proof, we need the notion of perfect splines.

A perfect spline s of degree k with (simple) knots $\zeta_1 < \dots < \zeta_r$ in $[a,b]$ is any function s of the form

$$s(x) = P(x) + c \sum_{i=0}^{r} (-)^i \int_{\zeta_i}^{\zeta_{i+1}} (x - y)_+^{k-1} dy \qquad (\zeta_0 := a,\ \zeta_{r+1} := b)$$

with $P \in \mathbb{P}_k :=$ polynomials of degree $< k$ and c some constant. In other words, such a function s is any k-th anti-derivative of an absolutely constant function with r sign changes or jumps in $[a,b]$. Their connection with the present topic stems from

Proposition 1 (Karlin [10]). *If s is a perfect spline of degree k with $n - k$ knots then*

$$\|s^{(k)}\| = \inf\{\|f^{(k)}\| : f \in \mathbb{L}_\infty^{(k)},\ f|_\tau = s|_\tau\} .$$

This proposition follows directly from the following lemma.

Lemma 1. *If $h \in \mathbb{L}_\infty^{(k)}$ is such that, for $a = \zeta_0 < \dots < \zeta_{r+1} = b$,*

$$\sigma_i h^{(k)} > 0 \text{ a.e. on } (\zeta_i, \zeta_{i+1}) \text{ for some } \sigma_i \in \{-1,1\},\ i=0,\dots,r ,$$

and $h|_\tau = 0$, then $r \geq n - k$. Further, if $r \leq n - k$, hence $r = n - k$, then

(3) $\tau_i < \zeta_i < \tau_{i+k}$, *all* i .

Assuming this lemma for the moment, we see that if, in Proposition 1, we had $\|s^{(k)}\| > \|f^{(k)}\|$ for some $f \in \mathbb{L}_\infty^{(k)}$ with $f|_\tau = s|_\tau$, then $h := s - f$ would satisfy the hypotheses of the lemma for some $r < n - k$, which would contradict the conclusion of the lemma.

Proof of Lemma 1: By Rolle's theorem, there must be points

$$t_1 < \dots < t_{n+1-k}$$

at which $h^{(k-1)}$ vanishes, and for which

(4) $\tau_i \leq t_i \leq \tau_{i+k-1}$, all i.

Then, each interval $[\zeta_i, \zeta_{i+1}]$ contains at most one of the t_j since, on each such interval, $h^{(k-1)}$ is strictly monotone, by assumption. Therefore, if n_j denotes the number of such intervals to which t_j belongs, then

$$n+1-k \leq \Sigma n_j \leq r+1 ,$$

or $n-k \leq r$. This also shows that, if $n-k=r$, then $n_j = 1$, all j, and so

$$\zeta_{i-1} < t_i < \zeta_i, \text{ all } i \text{ (except that, possibly, } \zeta_0 = t_1, \; t_{n+1-k} = \zeta_{r+1})$$

which, combined with (4) then implies (3). QED.

We also need

<u>Theorem</u> 1 (Karlin [10]). <u>If</u> $f \in \mathbb{L}_\infty^{(k)}$, <u>then there exists a perfect spline of degree</u> k <u>with</u> $< n-k$ <u>knots which agrees with</u> f <u>at</u> τ.

A simple proof can be found in [2].

We are now fully equipped for the attack on the optimal recovery scheme. By Karlin's theorem, the set

$$Q_\tau := \{q : q \text{ is a perfect spline of degree } k \text{ with } \leq n-k \text{ knots, } q|_\tau = 0\}$$

is not empty. By Lemma 1, each $q \in Q_\tau \backslash \{0\}$ has, in fact, exactly $n-k$ knots and does not vanish off τ (since otherwise it would have $n+1$ zeros yet only $n-k$ sign changes in its k-th derivative, contrary to the lemma). Therefore, if $f \in \mathbb{L}_\infty^{(k)}$ and $f|_\tau = 0$ and $x \notin \tau$, then $(f(x)/q(x))q$ is a well defined perfect spline of degree k with $n-k$ knots which agrees with f at the n points of τ and the additional point x, hence, Proposition 1 implies that

(5) $\|f^{(k)}\| \geq |f(x)/q(x)| \, \| q^{(k)} \|$.

It follows that

(6) $\sup\{|f(x)|/\|f^{(k)}\| : f \in \mathbb{L}_\infty^{(k)},\ f|_\tau = 0\} = |q(x)|/\|q^{(k)}\|$.

Since the left side is independent of q, this shows that Q_τ is the span of just one function, say of $\hat{q}$, normalized to have

$$\hat{q}^{(k)}(0^+) = 1 .$$

It also shows that

(7) $c(\tau) = \|\hat{q}\|$.

This gives a means of computing $c(\tau)$, but not quite yet the equality $\text{const}_\tau = c(\tau)$ nor the optimal recovery scheme. For this, let $\xi_1,\dots,\xi_{n-k}$ be the $n-k$ knots of $\hat{q}$ and consider

$$\begin{aligned}\$ &:= \text{splines of order } k \text{ with simple knots } \xi_1,\dots,\xi_{n-k}\\ &:= \mathbb{P}_{k,\xi} \cap C^{(k-2)}\\ &:= \{f \in C^{(k-2)} : f|_{(\xi_i,\xi_{i+1})} \in \mathbb{P}_k,\ \text{all } i\} \quad (\xi_0 = a, \xi_{n+1-k} = b).\end{aligned}$$

Theorem 2 (Micchelli, Rivlin & Winograd [15]). <u>The rule</u>

$$(\hat{S}f)_\tau = f|_\tau \text{ and } \hat{S}f \in \$, \text{ all } f \in \mathbb{L}_\infty^{(k)}$$

<u>defines a map</u> $\hat{S}$ (read "ess crown(ed)") <u>on</u> $\mathbb{L}_\infty^{(k)}$ <u>which is linear and an optimal recovery scheme with respect to</u> τ.

<u>Proof</u>. First, we prove that $\hat{S}$ is well defined. Since, by the lemma, $\tau_i < \xi_i < \tau_{i+k}$, we ask for matching $f^{(k-1)}$ at some point τ_i only when such $\tau_i \notin \xi$, i.e., when $s^{(k-1)}(\tau_i)$ makes sense for each $s \in \$$. Secondly, $\dim \$ = k + \text{number of polynomial pieces} - 1 = n$, therefore $\hat{S}$ is well defined if and only if

(8) $f|_\tau = 0$ and $f \in \$$ implies $f = 0$.

For this, we would like to use inequality (5), but, since $\$$ is not contained in $\mathbb{L}_\infty^{(k)}$ but only in the larger space

$$\mathbb{L}_{\infty,\xi}^{(k)} := \mathbb{L}_\infty^{(k-1)} \cap \bigcap_{i=0}^{n-k} \mathbb{L}_\infty^{(k)}(\xi_i,\xi_{i+1}) ,$$

we must first extend (5) to such spaces, with the convention that

$$\| f^{(k)} \| := \max_i \| f^{(k)} \big|_{(\zeta_i,\zeta_{i+1})} \|_\infty \quad \text{for} \quad f \in \mathbb{L}^{(k)}_{\infty,\xi} .$$

This requires the following slight strengthening of Lemma 1.

Lemma 1'. *If* $h \in \mathbb{L}^{(k)}_{\infty,\zeta}$ *for some* $a = \zeta_0 < \ldots < \zeta_{r+1} = b$ *and, for some* $\sigma \in \{-1,1\}$,

$$\sigma(-)^i h^{(k)} > 0 \quad \textit{a.e. on} \quad (\zeta_i,\zeta_{i+1}), \quad i = 0,\ldots,r ,$$

then $h\big|_\tau = 0$ *implies* $n - k \le r$.

Here, we mean by $h^{(k-1)}(x) = 0$ that $h^{(k-1)}(x^-)h^{(k-1)}(x^+) \le 0$, in case the number x occurs k-fold in τ.

Proof. Rolle's theorem gives again $n + 1 - k$ distinct zeros of $h^{(k-1)}$ which again consists of $r + 1$ strictly monotone pieces, but may fail to be continuous across the points $\zeta_1,\ldots,\zeta_r$. But, since $h^{(k)}$ alternates in sign, this latter fault can easily be remedied by appropriate local linear interpolation across



a small neighborhood of each discontinuity ζ_i without disturbing the other two properties and now $n - k \le r$ follows as before; Q.E.D.

This gives

Proposition 1'. *If* s *is a perfect spline with* $< n - k$ *knots, say with knots* $\zeta_1 < \ldots < \zeta_r$ *where* $r < n - k$, *and of* degree k, then even

$$\| s^{(k)} \| = \inf\{ \| f^{(k)} \| : f \in \mathbb{L}_{\infty,\zeta}^{(k)},\ f|_\tau = s|_\tau \} .$$

From this we conclude, with $\| \hat{q}^{(k)} \| = 1$, that

$$(5')\quad \| f^{(k)} \| \geq |f(x)/\hat{q}(x)| \quad \text{for} \quad x \notin \tau,\ f \in \mathbb{L}_{\infty,\xi}^{(k)},\ f|_\tau = 0 .$$

But this implies (8) since $\| f^{(k)} \| = 0$ for $f \in \$$, and so shows that $\hat{S}$ is well defined. Finally, (5') also implies that, for $f \in \mathbb{L}_\infty^{(k)}$,

$$\| f^{(k)} \| = \| (f - \hat{S}f)^{(k)} \| \geq |f(x) - \hat{S}f(x)| / |\hat{q}(x)|$$

or

$$(9)\quad |f(x) - \hat{S}f(x)| \leq |\hat{q}(x)| \, \| f^{(k)} \|$$

showing, with (2) and (7), that $\hat{S}$ is an optimal recovery scheme. This proves Theorem 2.

MRW actually insist that a recovery scheme S map into $\mathbb{L}_\infty^{(k)}$, hence they are not quite done at this point, since $\hat{S}$ only maps into $\mathbb{L}_\infty^{(k-1)}$. But, since $\$$ can be viewed as spline functions of degree k with double knots at the ξ_i, we can produce an element of $\mathbb{L}_\infty^{(k)}$ arbitrarily close to $\hat{S}f$ merely by pulling all these double knots apart ever so slightly. This shows that $\inf \text{const}_S = c(\tau)$ even if the inf is restricted to S mapping into $\mathbb{L}_\infty^{(k)}$ but now the inf is not attained apparently for $k > 1$.

3. THE ENVELOPE CONSTRUCTION

The preceding discussion allows a simple derivation of sharp estimates for the value of a function $f \in \mathbb{L}_\infty^{(k)}$ at some point x, given the vector $f|_\tau$ and a bound L on its k-th derivative on [a,b], as follows.

We are to construct the set

$$I_x := \{ f(x) : f \in F \}$$

with

$$F := \{f \in \mathbb{L}_\infty^{(k)} : f\big|_\tau = \alpha,\ \|f^{(k)}\| \le L\}$$

for some given α and L. If F is not empty, then I_x is a closed interval,

$$I_x =: [a_x, b_x]$$

say, since F is closed, convex and bounded and $[x] : f \mapsto f(x)$ is a continuous linear functional on $\mathbb{L}_\infty^{(k)}$. Assume that

$$F^o := \{f \in F : \|f^{(k)}\| < L\} \ne \phi .$$

Then

(10) $(a_x, b_x) = [x]F^o$.

Karlin's theorem then implies that, for $x \notin \tau$,

$$Q_x := \{q : q \text{ is perfect spline of degree } k \text{ with } \le n-k \text{ knots},\ q\big|_\tau = \alpha,\ q(x) = a_x\}$$

is not empty. Further, $Q_x \subseteq F$, since, by definition of a_x, there exists $f \in F$ with $f(x) = a_x$ while each $q \in Q_x$ agrees with such an f at the $n+k+1$ points τ and x, therefore $\|q^{(k)}\| \le \|f^{(k)}\| \le L$, by the proposition. On the other hand, $Q_x \cap F^o = \phi$, since, if $q \in Q_x \cap F^o$, then $a_x = q(x) \in (a_x, b_x)$, by (10), a contradiction.

It follows that

$$\text{for all } y,\ a_y \le q(y) \le b_y .$$

But, for any $g \in F^o$, $h := g - q$ has $\le n-k$ sign changes in its k-th derivative and vanishes at τ, therefore q has exactly $n-k$ knots and h does not vanish off τ, by Lemma 1. Hence, if $\tau_{i+1}, \ldots, \tau_{i+j}$ are all the points of τ between x and $y \notin \tau$, then

$$(-)^j (g(y) - q(y)) > 0 .$$

This shows that

$$q(y) = \begin{Bmatrix} a_y \\ b_y \end{Bmatrix} \text{ if the interval between } x \text{ and } y \text{ contains an } \begin{Bmatrix} \text{even} \\ \text{odd} \end{Bmatrix}$$

number of points of τ. It follows that Q_x contains exactly one function and that this function supplies half of the entire boundary of

(11) $\{(x,f(x)) : x \in [a,b],\ f \in F\}$,

the other half being supplied by the perfect spline p of degree k with n-k knots for which $p|_\tau = \alpha$ and $p(x) = b_x$.

We note the curious corollary that the perfect spline s of Karlin's theorem is unique if there exists g which agrees with f at all points of τ but one and for which $\|g^{(k)}\| < \|s^{(k)}\|$, i.e., if at least one of the interpolation constraints is active. Put differently, it says that if there are two different perfect splines $s, \hat{s}$ of degree k with $\leq n - k$ knots for which $s|_\tau = \hat{s}|_\tau = \alpha$ and which agree at some $x \notin \tau$, then

$$\|s^{(k)}\| = \|\hat{s}^{(k)}\| = L_\alpha := \min\{\|f^{(k)}\| : f \in \mathbb{L}_\infty^{(k)},\ f|_\tau = \alpha\} .$$

Let now q be the half of the envelope of (11) with $q^{(k)}(0^+) = L$, and let p be the one with $p^{(k)}(0^+) = -L$. Gaffney and Powell [8] choose

$$S_L\alpha := (p + q)/2$$

as a good interpolant, its graph being clearly the center of (11). Since p and q are uniquely defined for $L > L_\alpha$ by the requirement that they are perfect splines of degree k with n - k knots, equal to α at τ and to a_x, resp. b_x at x, they are necessarily continuous functions of L and α in that range. In particular, with $q =: q_{L,\alpha}$, we have $q_{L,\alpha} = Lq_{1,\alpha/L}$ and $q_{1,\alpha/L} \to q_{1,0} = \hat{q}$ as $L \to \infty$, and, similarly, $p_{L,\alpha/L} \to -\hat{q}$ as $L \to \infty$. This shows that, with $u_1 < \ldots < u_{n-k}$ the knots of q and $v_1 < \ldots < v_{n-k}$ the knots of p,

$$\lim_{L\to\infty} u_i = \lim_{L\to\infty} v_i = \xi_i, \quad i = 1,\ldots,n - k .$$

In particular, for large enough L, the sequence

$$0 = \xi_{0+}, \xi_{1-}, \xi_{1+}, \ldots, \xi_{n-k-}, \xi_{n-k+}, \xi_{n-k+1-} = 1$$

with

$$\xi_{i-} := \min\{u_i, v_i\},\ \xi_{i+} := \max\{u_i, v_i\}$$

is nondecreasing and

$$(S_L\alpha)^{(k)} = \begin{cases} 0 & \text{on} \quad (\xi_{i+},\xi_{i+1-}) \\ \pm 2L & \text{on} \quad (\xi_{i-},\xi_{i+}) \end{cases}$$

hence

$$\| (S_L\alpha)^{(k)} \|_1 \leq 2L \| u - v \|_1 \xrightarrow[L\to\infty]{} 0 \ .$$

This shows that $S_L\alpha$ converges to an element of $, while

$$S_L\alpha\big|_\tau = \alpha \quad \text{for all} \quad L \ ,$$

therefore, $S_L\alpha$ converges to the optimal interpolant for the data.

It was in this way that Gaffney and Powell [8] constructed, quite independently from Micchelli, Rivlin and Winograd [15], the same optimal recovery or interpolation scheme $\overset{*}{S}$.

The problem of constructing the set I_x was posed originally in the basic paper by Golomb and Weinberger [9], although they gave detailed attention to such problems only when the (semi)norm involved comes from an inner product. Micchelli and Miranker [14] solved the problem posed at the beginning of this section in the sense that they correctly described the boundary of (11) as being given by just two perfect splines of degree k, each with n - k knots, and with their k-th derivative equal to L in absolute value. In fact, Micchelli and Miranker consider the slightly more general situation where $f^{(k)}$ is only known to map [a,b] into some interval [m,M]. They state that these matters could be proved along the lines used by Burchard [6] to solve a related restricted moment problem and refer specifically to Karlin and Studden [11, VIII, §3] for requisite facts concerning principal representations of interior points of moment spaces. Of course, these facts go back to Krein [12]. Quite independently, Gaffney and Powell [8] also solved this problem, with the proofs in Gaffney's thesis based on Chebyshev type inequalities as found in Karlin and Studden [11, VIII, §8] and adapted by him to weak Chebyshev systems.

4. THE CONSTRUCTION OF NORM PRESERVING EXTENSIONS TO ALL OF $\mathbb{L}_1$

Both the optimal recovery scheme $\overset{*}{S}$ of Section 2 and the envelope of Section 3 require the construction of an absolutely constant function h with no more than a specified number of jumps which provides an integral representation of a linear functional given on a linear space of spline functions.

In the optimal recovery, we are to construct a perfect spline $\hat{s}$ of degree k with n - k knots and with $\|\hat{s}^{(k)}\| = 1$ which vanishes at the given n-point sequence τ. Let $(M_i)_1^n$ be the sequence of B-splines of order k with knot sequence τ, each normalized to have unit integral, i.e.,

(12) $M_i(x) := M_{i,k,\tau}(x) := k[\tau_i,\ldots,\tau_{i+k}](\cdot - x)_+^{k-1}$,

with $[\tau_i,\ldots,\tau_{i+k}]f$ the k-th divided difference of the function f at the points $\tau_i,\ldots,\tau_{i+k}$. Then, from Taylor's expansion with integral remainder,

$$[\tau_i,\ldots,\tau_{i+k}]f = \int_{\tau_i}^{\tau_{i+k}} M_{i,k,\tau}(x) f^{(k)}(x)\,dx/k! \text{ for all } f \in \mathbb{L}_\infty^{(k)}.$$

The points $\xi = (\xi_i)_1^{n-k}$ are therefore characterized by the requirement that the function

$$h_\xi(x) := \operatorname{sign} \prod_{i=1}^{n-k} (x - \xi_i) = \pm\hat{s}^{(k)}(x)$$

be orthogonal to each of the n - k functions $M_1,\ldots,M_{n-k}$.

Before considering the computational details of determining ξ from this orthogonality condition, I want to comment on the fact that this is a problem of representing or extending a linear functional on some subspace of $\mathbb{L}_1$ and is therefore closely related to the problem of computing the norm of a linear functional on some subspace of $\mathbb{L}_1$. This is also explored in a forthcoming paper by Micchelli [13].

Indeed, if T is a linear subspace of $\mathbb{L}_1$ of dimension n + 1 - k, and λ is a linear functional on T, we might ask for ξ and α so that

(13) $\alpha \int h_\xi g = \lambda g$ for all $g \in T$.

But then, in particular, h_ξ is orthogonal to ker λ, a subspace of dimension n - k. Conversely, if we have already found h_ξ orthogonal to ker λ then there will be exactly one α so that αh_ξ

represents λ on T in the sense of (13), unless h_ξ is even orthogonal to all of T. But this latter event cannot happen in case T is weak Chebyshev since h_ξ has only n - k jumps.

It is clear that any such representation h_ξ for λ produces an upper bound for the norm of λ. In fact, $\|\lambda\| = |\alpha|$ in case T is weak Chebyshev. This is actually how I became interested two years ago in the numerical construction of representers of linear functionals [3], [4]. I was interested in computing, or at least closely estimating, the norm of certain linear functionals on certain spline subspaces in $\mathbb{L}_1$. In a way, this is a trivial problem, viz. the maximization of a linear function over a finite dimensional compact convex set, and there was the feeling that there ought to be special methods available. Perhaps some reader can steer me towards such methods. I found, for the particular cases of concern to me in which T was always weak Chebyshev, nothing more effective for calculating $\|\lambda\|$ than to construct such a representation (13).

Finally, the envelope construction corresponds to the slightly different situation where dim T = n - k, $\lambda \in T'$ and α with $|\alpha| > \|\lambda\|$ is prescribed and one seeks ξ so that again (13) holds. We have now one less condition to satisfy but also one less parameter to do it with.

5. CONSTRUCTION OF THE KNOTS FOR THE OPTIMAL RECOVERY SCHEME

As we saw in the preceding section, the knots $\xi = (\xi_i)_1^{n-k}$ for the optimal recovery scheme are the solution to the following problem. We are given $(\tau_i)_1^n$ nondecreasing, with $\tau_i < \tau_{i+k}$, all i, and n > k, and wish to construct $\xi_0 < \ldots < \xi_r$ with

$$r := n - k + 1$$

and with $\xi_0 = a := \tau_1$, $\xi_r = b := \tau_n$ so that

$$(14)\quad \sum_{j=1}^{r} \beta_j \int_{\xi_{j-1}}^{\xi_j} M_{i,k} = 0, \quad i = 1,\ldots,n-k$$

while also

$$(15)\quad \beta_j + \beta_{j+1} = 0, \quad j = 1,\ldots,n-k\,.$$

Extend τ by

$$\tau_n =: \tau_{n+1} =: \tau_{n+2} =: \dots =: \tau_{n+k} .$$

Then one verifies easily that

$$\int_a^x M_{i,k} = \sum_{m \geq i} N_{m,k+1}(x) \quad \text{for} \quad x \geq a$$

with

$$N_{m,k+1} := \frac{\tau_{m+k+1} - \tau_m}{k+1} M_{m,k+1}, \quad \text{all} \quad m .$$

Therefore, (14) is equivalent to

$$\sum_{j=1}^{r} \beta_j \sum_{m \geq i} (N_{m,k+1}(\xi_j) - N_{m,k+1}(\xi_{j-1})) = 0, \quad i = 1,\dots,n-k ,$$

or, on subtracting equation i from equation i - 1 for i = 2,...,n-k,

$$\sum_{j=1}^{r} \beta_j (N_{i,k+1}(\xi_j) - N_{i,k+1}(\xi_{j-1})) = 0, \; i = 1,\dots,n-k-1$$

$$\sum_{j=1}^{r} \beta_j \sum_{m \geq n-k} (N_{m,k+1}(\xi_j) - N_{m,k+1}(\xi_{j-1})) = 0 .$$

Using the fact that $N_{m,k+1}(\xi_0) = 0$ for $m \geq 1$, this can also be written

$$\sum_{j=1}^{n-k} (\beta_j - \beta_{j+1}) N_{i,k+1}(\xi_j) = -\beta_r N_{i,k+1}(\xi_r), \; i = 1,\dots,n-k-1$$

$$\sum_{j=1}^{n-k} (\beta_j - \beta_{j+1}) \sum_{m \geq n-k} N_{m,k+1}(\xi_j) = -\beta_r \sum_{m \geq n-k} N_{m,k+1}(\xi_r) .$$

Since $N_{i,k+1}(\xi_r) = 0$ for $i < n-k$, while $\sum_{m \geq n-k} N_{m,k+1}(\xi_r) = 1$, the right side becomes simply $(0,\dots,0,-\beta_r)^T$.

Choose now $\beta_r = -1$ to make things definite. Then $\beta_j = (-)^{r-j-1} = (-)^{n-k-j}$ by (15), and (19) and (15) are seen to be equivalent to

(16) $F(\xi) = 0$

with $F : \mathbb{R}^{n-k} \to \mathbb{R}^{n-k}$ given by

$$(17)\quad F(\xi)_i := \begin{cases} a_i & , i < n - k \\ \sum_{m=n-k}^{n} a_m, & i = n - k \end{cases}$$

where

$$(18)\quad a_i := \begin{cases} \sum_{j=1}^{n-k} (-)^{n-k-j} N_{i,k+1}(\xi_j), & i = 1,\dots,n-1 \\ -1/2 & , i = n\ . \end{cases}$$

We solve (16) by Newton's method. From the current guess ξ, we compute a new guess $\xi^* = \xi + \delta\xi$, with $\delta\xi$ the solution to the linear system

$$(19)\quad F'(\xi)\delta\xi = -F(\xi)\ .$$

Since $N'_{i,k+1} = M_{i,k} - M_{i+1,k}$, addition of equation i to equation i-1, $i = n - k,\dots,2$, in (19) produces the equivalent linear system

$$\sum_{j=1}^{n-k} \sum_{m=i}^{n-1} (M_{m,k} - M_{m+1,k})(\xi_j)\ (-)^{n-k-j}\delta\xi_j = -\sum_{m=i}^{n} a_m,\ i=1,\dots,n-k.$$

But, since $\sum_{i}^{n-1} (M_{m,k} - M_{m+1,k})(t) = M_{i,k}(t) - M_{n,k}(t) = M_{i,k}(t)$ for $t \le b$, this shows that (19) is equivalent to the linear system

$$(20)\quad Cx = d$$

with

$$(21)\quad (-)^{n-k-j}\delta\xi_j = x_j,\quad d_i = \Big(\sum_{m=i}^{n} -a_m\Big)(\tau_{i+k} - \tau_i)/k,\quad c_{ij} = N_{i,k}(\xi_j)$$

$$i,j = 1,\dots,n-k.$$

The matrix C is totally positive and (2k - 1)-banded, hence can be stored in 2k - 1 bands of length n - k each, and can be factored cheaply and reliably within these bands by Gauss elimination <u>without</u> pivoting (see [5]).

The iteration is carried out in the program SPLOPT below, starting with the initial guess

$$(22)\quad \xi_i = (\tau_{i+1} + \dots + \tau_{i+k-1})/(k - 1),\ i = 1,\dots,n-k\ .$$

A first version of the program was equipped to carry out Modified Newton iteration: ξ^* is computed as the first vector in the sequence

$$\xi + 2^{-h}\delta\xi, \quad h = 0,1,2,\ldots$$

for which $\| F(\xi + 2^{-h}\delta\xi)\|_2 < \| F(\xi)\|_2$. But, in all examples tried, the initial guess (22) was apparently sufficiently close to the solution to have always $h = 0$, i.e., $\| F(\xi + \delta\xi)\|_2 < \|F(\xi)\|_2$. In fact, the termination criterion

$$\| \delta\xi\|_\infty \le 10^{-6}(\tau_n - \tau_1)/(n - k)$$

was usually reached in three or four clearly quadratically converging iterations. For this reason, the program SPLOPT below carries out simple Newton iteration. It would be nice to prove that Newton iteration, starting from (22), necessarily converges. But, such a proof would necessarily have to be in control of the norm of the inverse of the matrix $F'(\xi)$, hence in control of the norm of the inverse of $C = (N_i(\xi_j))$ as a function of ξ. Good estimates for $\| (N_i(\xi_j))^{-1}\|$ have been searched for in the past by some who were interested in bounding (the error in) spline interpolation, but without much success. E.g., the simple conjecture that, for the initial guess (22), $\| (N_i(\xi_j))^{-1}\|_1 \le \text{const}_k$ independent of ξ has been proved so far only for $k \le 4$.

```
      SUBROUTINE SPLOPT ( TAU, N, K, SCRTCH, T, IFLAG )
COMPUTES THE KNOTS  T  FOR THE OPTIMAL RECOVERY SCHEME OF ORDER  K
C  FOR DATA AT  TAU (I),I=1,...., N .  TAU  MUST BE STRICTLY INCREASING.
C  SEE TEXT FOR DEFINITION OF VARIABLES AND METHOD USED.
C   IFLAG  =  1 OR 2 DEPENDING ON WHETHER OR NOT  T  WAS CONSTRUCTED.
C     DIMENSION SCRTCH((N-K)*(2*K+3)+5*K+3), T(N+K)
      DIMENSION TAU(N),SCRTCH(1),T(1)
      DATA NEWTMX,TOLRTE / 10,.000001/
      NMK = N-K
      IF (NMK)                                 1,56,2
    1 PRINT 601,N,K
  601 FORMAT(13H ARGUMENT N =,I4,29H IN  SPLOPT  IS LESS THAN K =,I3)
                                               GO TO 999
    2 IF (K .GT. 2)                            GO TO 3
      PRINT 602,K
  602 FORMAT(13H ARGUMENT K =,I3,27H IN  SPLOPT  IS LESS THAN 3)
                                               GO TO 999
    3 NMKM1 = NMK - 1
      FLOATK = K
      KPK = K+K
      KP1 = K+1
      KPKP1 = K+KP1
      KM1 = K-1
      KPKM1 = K+KM1
      KPN = K+N
      SIGNST = -1.
      IF (NMK .GT. (NMK/2)*2)  SIGNST = 1.
```

```
C   SCRTCH(I) = TAU-EXTENDED(I), I=1,...,N+K+K
      NX = N+KPKP1
C   SCRTCH(I+NX) = XI(I),I=0,...,N-K+1
      NA = NX + NMK + 1
C   SCRTCH(I+NA) = -A(I), I=1,...,N
      ND = NA + N
C   SCRTCH(I+ND) = X(I) OR D(I), I=1,...,N-K
      NV = ND + NMK
C   SCRTCH(I+NV) = VNIKX(I),I=1,...,K+1
      NC = NV + KP1
C   SCRTCH((J-1)*(N-K)+I + NC) = CHAT(I,J),I=1,...,N-K,J=1,...,2*K-1
      LENGCH = NMK*KPKM1
C   EXTEND  TAU  TO A KNOT SEQUENCE AND STORE IN SCRTCH.
      DO 5 J=1,K
         SCRTCH(J) = TAU(1)
    5    SCRTCH(KPN+J) = TAU(N)
      DO 6 J=1,N
    6    SCRTCH(K+J) = TAU(J)
C   FIRST GUESS FOR  SCRTCH (.+NX)  =  XI .
      SCRTCH(NX) = TAU(1)
      SCRTCH(NMK+1+NX) = TAU(N)
      DO 10 J=1,NMK
         SUM = 0.
         DO 9 L=1,KM1
    9       SUM = SUM + TAU(J+L)
   10    SCRTCH(J+NX) = SUM/KM1
C   LAST ENTRY OF  SCRTCH (.+NA)  =  - A  IS ALWAYS ...
      SCRTCH(N+NA) = .5
C   START NEWTON ITERATION.
      NEWTON = 1
      TOL = TOLRTE*(TAU(N) - TAU(1))/NMK
C   START NEWTON STEP
COMPUTE THE 2K-1 BANDS OF THE MATRIX C AND STORE IN SCRTCH(.+NC),
C   AND COMPUTE THE VECTOR  SCRTCH(.+NA) = -A.
   20 DO 21 I=1,LENGCH
   21    SCRTCH(I+NC) = 0.
      DO 22 I=2,N
   22    SCRTCH(I-1+NA) = 0.
      SIGN = SIGNST
      ILEFT = KP1
      DO 28 J=1,NMK
         XIJ = SCRTCH(J+NX)
   23       IF (XIJ .LT. SCRTCH(ILEFT+1))   GO TO 25
            ILEFT = ILEFT + 1
            IF (ILEFT .LT. KPN)        GO TO 23
            ILEFT = ILEFT - 1
   25    CALL BSPLVN(SCRTCH,K,1,XIJ,ILEFT,SCRTCH(1+NV))
         ID = MAX0(0,ILEFT-KPK)
         INDEX = NC+(J-ID+KM1)*NMK+ID
         LLMAX = MIN0(K,NMK-ID)
         LLMIN = 1 - MIN0(0,ILEFT-KPK)
         DO 26 LL=LLMIN,LLMAX
            INDEX = INDEX - NMKM1
   26       SCRTCH(INDEX) = SCRTCH(LL+NV)
         CALL BSPLVN(SCRTCH,KP1,2,XIJ,ILEFT,SCRTCH(1+NV))
         ID = MAX0(0,ILEFT-KPKP1)
         LLMIN = 1 - MIN0(0,ILEFT-KPKP1)
         DO 27 LL=LLMIN,KP1
            ID = ID + 1
   27       SCRTCH(ID+NA) = SCRTCH(ID+NA) - SIGN*SCRTCH(LL+NV)
   28    SIGN = -SIGN
```

```
      CALL BANFAC(SCRTCH(1+NC),NMK,NMK,KPKM1,K,IFLAG)
                                             GO TO (45,44),IFLAG
   44 PRINT 644
  644 FORMAT(32H C IN  SPLOPT  IS NOT INVERTIBLE)
                                             RETURN
COMPUTE  SCRTCH (.+ND) =  D  FROM  SCRTCH (.+NA) = - A .
   45 DO 46 I=N,2,-1
   46    SCRTCH(I-1+NA) = SCRTCH(I-1+NA) + SCRTCH(I+NA)
      DO 49 I=1,NMK
   49    SCRTCH(I+ND) = SCRTCH(I+NA)*(TAU(I+K)-TAU(I))/FLOATK
COMPUTE  SCRTCH (.+ND) =  X .
      CALL BANSUB(SCRTCH(1+NC),NMK,NMK,KPKM1,K,SCRTCH(1+ND))
COMPUTE  SCRTCH (.+ND) = CHANGE IN  XI . MODIFY, IF NECESSARY, TO
C  PREVENT NEW  XI  FROM MOVING MORE THAN 1/3 OF THE WAY TO ITS
C  NEIGHBORS. THEN ADD TO  XI  TO OBTAIN NEW  XI  IN SCRTCH(.+NX).
      DELMAX = 0.
      SIGN = SIGNST
      DO 53 I=1,NMK
         DEL = SIGN*SCRTCH(I+ND)
         DELMAX = AMAX1(DELMAX,ABS(DEL))
         IF (DEL .GT. 0.)                    GO TO 51
         DEL = AMAX1(DEL,(SCRTCH(I-1+NX)-SCRTCH(I+NX))/3.)
                                             GO TO 52
   51    DEL = AMIN1(DEL,(SCRTCH(I+1+NX)-SCRTCH(I+NX))/3.)
   52    SIGN = -SIGN
   53    SCRTCH(I+NX) = SCRTCH(I+NX) + DEL
CALL IT A DAY IN CASE CHANGE IN  XI  WAS SMALL ENOUGH OR TOO MANY
C  STEPS WERE TAKEN.
      IF (DELMAX .LT. TOL)                   GO TO 54
      NEWTON = NEWTON + 1
      IF (NEWTON .LE. NEWTMX)                GO TO 20
      PRINT 653,NEWTMX
  653 FORMAT(33H NO CONVERGENCE IN  SPLOPT  AFTER,I3,14H NEWTON STEPS.)
   54 DO 55 I=1,NMK
   55    T(K+I) = SCRTCH(I+NX)
   56 DO 57 I=1,K
         T(I) = TAU(1)
   57    T(N+I) = TAU(N)
                                             RETURN
  999 IFLAG = 2
                                             RETURN
      END
```

The subroutine SPLOPT has input TAU(i) = τ_i, i = 1,...,n, assumed to be nondecreasing and to satisfy $\tau_i < \tau_{i+k}$, all i, the integer N = n and the desired order k in K. The routine needs a work array SCRTCH, of size $\geq$ (n - k) (2k + 3) + 5k + 3; n(2k + 3) is more than enough. The routine has output T(i) = t_i, i = 1,...,n + k, the knot sequence for the optimal recovery scheme, in case IFLAG = 1. For IFLAG = 2, something went wrong.

The routine uses the subroutine BSPLVN for the evaluation of all B-splines of a given order on a given knot sequence which do not vanish at a given point. This routine, and others for dealing computationally with splines and B-splines, can be found in [1]. For completeness, we also list here the subroutines BANFAC and BANSUB, used in SPLOPT for the solution of the banded system (20).

```
      SUBROUTINE BANFAC ( A, NROW, N, NDIAG, MIDDLE, IFLAG )
      DIMENSION A(NROW,NDIAG)
      IFLAG = 1
      ILO = MIDDLE - 1
      IF (ILO)                                  999,10,19
   10 DO 11 I=1,N
         IF (A(I,1))                            11,999,11
   11    CONTINUE
                                                RETURN
   19 IHI = NDIAG - MIDDLE
      IF (IHI)                                  999,20,29
   20 DO 25 I=1,N
         IF (A(I,MIDDLE))                       21,999,21
   21    JMAX = MINO(ILO,N-I)
         IF (JMAX)                              25,25,22
   22    DO 23 J=1,JMAX
   23       A(I+J,MIDDLE-J) = A(I+J,MIDDLE-J)/A(I,MIDDLE)
   25    CONTINUE
                                                RETURN
   29 DO 50 I=1,N
         DIAG = A(I,MIDDLE)
         IF (DIAG)                              31,999,31
   31    JMAX = MINO(ILO,N-I)
         IF (JMAX)                              50,50,32
   32    KMAX = MINO(IHI,N-I)
         DO 33 J=1,JMAX
            IPJ = I+J
            MMJ = MIDDLE-J
            A(IPJ,MMJ) = A(IPJ,MMJ)/DIAG
            DO 33 K=1,KMAX
   33          A(IPJ,MMJ+K) = A(IPJ,MMJ+K) - A(IPJ,MMJ)*A(I,MIDDLE+K)
   50    CONTINUE
                                                RETURN
  999 IFLAG = 2
                                                RETURN
      END
```

BANFAC factors an $N \times N$ band matrix C whose NDIAG bands are contained in the columns of the NROW $\times$ NDIAG array A, with the MIDDLE column containing the main diagonal of C. It uses Gauss elimination without pivoting and stores the factors in A.

```
      SUBROUTINE BANSUB ( A, NROW, N, NDIAG, MIDDLE, B )
      DIMENSION A(NROW,NDIAG),B(N)
      IF (N .EQ. 1)                                     GO TO 21
      ILO = MIDDLE - 1
      IF (ILO)                                          21,21,11
   11 DO 19 I=2,N
         JMAX = MINO(I-1,ILO)
         DO 19 J=1,JMAX
   19       B(I) = B(I) - B(I-J)*A(I,MIDDLE-J)
C
   21 I = N
      IHI = NDIAG-MIDDLE
      DO 29 II=1,N
         JMAX = MINO(N-I,IHI)
         IF (JMAX)                                      28,28,22
   22    DO 25 J=1,JMAX
   25       B(I) = B(I) - B(I+J)*A(I,MIDDLE+J)
   28    B(I) = B(I)/A(I,MIDDLE)
   29    I = I - 1
      END
```

BANSUB then uses the factorization of C into the product of a lower and an upper triangular matrix computed in BANFAC to solve the equation $Cx = b$ for given b (input in B) by forward and back substitution. The solution x is contained in B, on output.

6. CONSTRUCTION OF THE OPTIMAL INTERPOLANT

With the break points $\xi_1 < \ldots < \xi_{n-k}$ for the optimal interpolant $\hat{S}f$ determined from τ in SPLOPT, it remains to compute $\hat{S}f$. This we propose to do by determining its B-spline coefficients. SPLOPT has produced the knot sequence $\underline{\underline{t}} = (t_i)_1^{n+k}$, with

$$t_1 = \ldots = t_k = \tau_1, \; t_{k+i} = \xi_i, \; i = 1,\ldots,n-k,$$

$$t_{n+1} = \ldots \; t_{n+k} = \tau_n .$$

Let $(N_i)_1^n$ be the corresponding sequence of normalized B-splines of order k, i.e.,

$$N_i(x) := N_{i,k,\underline{\underline{t}}}(x) := (t_{i+k}-t_i)[t_i,\ldots,t_{i+k}](\cdot-x)_+^{k-1}, \text{ all } i.$$

Then, according to Curry & Schoenberg [7], every piecewise polynomial function of order k on $[a,b] := [\tau_1,\tau_n]$, with $k - 2$ continuous derivatives and break points $\xi_1,\ldots,\xi_{n-k}$, i.e., every spline of order k on $[a,b]$ with knot sequence $\underline{\underline{t}}$, has a unique representation as a linear combination of the n functions $N_1,\ldots,N_n$. Therefore

$$\hat{S}f = \sum_{i=1}^{n} a_i N_i$$

with $a = (a_i)_1^n$ the solution of the linear system

$$(23) \quad \sum_{j=1}^{n} N_j(\tau_i)a_j = f(\tau_i),\ i = 1,\ldots,n .$$

In case τ is strictly increasing, - the only case considered here, - Lemma 1 implies that $t_i < \tau_i < t_{i+k}$, all i, which, together with the fact that N_j vanishes outside the interval $[t_j,t_{j+k}]$, all j, shows that the coefficient matrix of (23) is $2k - 1$ banded. Since the coefficient matrix is also totally positive, we can therefore (see [5]) solve (23) by Gauss elimination without pivoting and within the $2k - 1$ bands required for the storage of the nonzero entries of the matrix. The following subroutine SPLINT generates the linear system (23), given on <u>input</u> the arrays TAU(i) = τ_i, FTAU(i) = $f(\tau_i)$, i = 1,...,N, T(i) = t_i, i = 1,...,N + K and the numbers N and K. The system is then solved, using BANFAC and BANSUB given in Section 5, and using a working array Q, of size N(2K - 1). The <u>output</u> consists of the B-coefficients a_i = BCOEF(i), i = 1,...,N, in case IFLAG = 1. If IFLAG = 2, then the linear system (23) was not invertible.

```
      SUBROUTINE SPLINT ( TAU, FTAU, T, N, K, Q, BCOEF, IFLAG )
C  SPLINT  PRODUCES THE B-SPLINE COEFF.S  BCOEF  OF THE SPLINE OF ORDER
C  K  WITH KNOTS  T (I), I=1,..., N + K , WHICH TAKES ON THE VALUE
C  FTAU (I) AT  TAU (I), I=1,..., N .
C  TAU  IS ASSUMED TO BE STRICTLY INCREASING.
C  SEE TEXT FOR DESCRIPTION OF VARIABLES AND METHOD.
C  DIMENSION T(N+K),Q(N,2*K-1)
      DIMENSION TAU(N), FTAU(N), T(1), Q(N,1), BCOEF(N)
      NP1 = N + 1
      KPKM1 = 2*K - 1
      ILEFT = K
      DO 30 I=1,N
         TAUI = TAU(I)
         ILP1MX = MINO(I+K,NP1)
         DO 13 J=1,KPKM1
   13       Q(I,J) = 0.
         ILEFT = MAXO(ILEFT,I)
         IF (TAUI .LT. T(ILEFT))          GO TO 998
   15       IF (TAUI .LT. T(ILEFT+1))     GO TO 16
            ILEFT = ILEFT + 1
            IF (ILEFT .LT. ILP1MX)        GO TO 15
         ILEFT = ILEFT - 1
         IF (TAUI .GT. T(ILEFT+1))        GO TO 998
   16    CALL BSPLVN ( T, K, 1, TAUI, ILEFT, BCOEF )
C        NOTE THAT  BCOEF  IS USED HERE FOR TEMP.STORAGE.
         L = ILEFT - I
         DO 30 J=1,K
            L = L+1
   30       Q(I,L) = BCOEF(J)
      NP2MK = N+2-K
      CALL BANFAC ( Q, N, N, KPKM1, K, IFLAG )
                                          GO TO (40,999), IFLAG
   40 DO 41 I=1,N
   41    BCOEF(I) = FTAU(I)
      CALL BANSUB ( Q, N, N, KPKM1, K, BCOEF )
                                          RETURN
  998 IFLAG = 2
  999 PRINT 699
  699 FORMAT(41H LINEAR SYSTEM IN  SPLINT  NOT INVERTIBLE)
                                          RETURN
      END
```

Note that [1] contains programs which might facilitate subsequent use of the optimal interpolant determined in this way via SPLOPT and SPLINT.

Finally, the linear system (23) can be generated in $O(nk^2)$ operations and, because of the band structure, can be solved in $O(nk)$ operations. The linear system (20), to be generated and solved at each Newton step for finding ξ, is of similar nature (with $n - k$ rather than n equations and a coefficient matrix which is the transposed of the kind of matrix appearing in (23)) hence requires a similar effort for its construction and solution. Therefore, if it takes indeed only three to four Newton iterations to find ξ to sufficient accuracy, then it takes only four to five times as much work to construct the optimal interpolant rather than any spline interpolant *to the same data*. Also, the total effort is only $O(nk^2)$ which, for large n, compares very favorably with such interpolation schemes as polynomial interpolation which takes $O(n^2)$ operations, or more general schemes which take as much as $O(n^3)$ operations.

REFERENCES

1. C. de Boor, Package for calculating with B-splines, MRC TSR 1333 (1973); SIAM J. Numer. Anal. (to appear).

2. C. de Boor, A remark concerning perfect splines, Bull. Amer. Math. Soc. 80 (1974) 724-727.

3. C. de Boor, A smooth and local interpolant with 'small' k-th derivative, in "Numerical solutions of boundary value problems for ordinary differential equations", A. K. Aziz ed., Academic Press, New York, 1975, 177-197.

4. C. de Boor, On local linear functionals which vanish at all B-splines but one, in "Theory of Approximation with Applications", A. G. Law & B. N. Sahney eds., Academic Press, New York, 1976, 120-145.

5. C. de Boor & A. Pinkus, Backward error analysis for totally positive linear systems, MRC TSR 1620, Jan. 1976; Numer. Math. (to appear).

6. H. G. Burchard, Interpolation and approximation by generalized convex functions, Ph.D. Thesis, Purdue U., Lafayette, IN., 1968.

7. H. B. Curry and I. J. Schoenberg, On Pólya frequency functions. IV: The fundamental spline functions and their limits, J. d'Analyse Math. 17 (1966) 71-107.

8. P. W. Gaffney & M. J. D. Powell, Optimal interpolation, in "Numerical Analysis", G. A. Watson ed., Lecture Notes in Math. Vol. 506, Springer, Heidelberg, 1976.

9. M. Golomb and H. F. Weinberger, Optimal approximation and error bounds, in "On numerical approximation", R. E. Langer ed., Univ. Wisconsin Press, Madison, WI., 1959, 117-190.

10. S. Karlin, Some variational problems on certain Sobolev spaces and perfect splines, Bull. Amer. Math. Soc. 79 (1973) 124-128.

11. S. Karlin and W. J. Studden, Tchebycheff systems: with applications in analysis and statistics, Interscience Publishers, John Wiley, New York, 1966.

12. M. G. Krein, The ideas of P. L. Chebyshev and A. A. Markov in the theory of limiting values of integrals and their further development, Uspekhi Mat. Nauk (n.S.) 6 (1951), 3-120; Engl. Transl.: Amer. Math. Soc. Transl. Series 2, 12 (1959) 1-122.

13. C. A. Micchelli, Best L^1-approximation by weak Chebyshev systems and the uniqueness of interpolating perfect splines, J. Approximation Theory (to appear).

14. C. A. Micchelli & W. L. Miranker, High order search methods for finding roots, J. ACM 22 (1975) 51-60.

15. C. A. Micchelli, T. J. Rivlin and S. Winograd, The optimal recovery of smooth functions, Numer. Math. 26 (1976) 279-285.

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: 1706
2. GOVT ACCESSION NO.
3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle): COMPUTATIONAL ASPECTS OF OPTIMAL RECOVERY
5. TYPE OF REPORT & PERIOD COVERED: Technical Summary Report, - no specific reporting period
6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s): Carl de Boor
8. CONTRACT OR GRANT NUMBER(s): DAAG29-75-C-0024
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Mathematics Research Center, University of Wisconsin, 610 Walnut Street, Madison, Wisconsin 53706
10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
11. CONTROLLING OFFICE NAME AND ADDRESS: U. S. Army Research Office, P.O. Box 12211, Research Triangle Park, North Carolina 27709
12. REPORT DATE: December 1976
13. NUMBER OF PAGES: 23
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): 26p.
15. SECURITY CLASS. (of this report): UNCLASSIFIED
15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report): MRC-TSR-1706

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Optimal recovery
Perfect splines
Spline interpolation
Extension of linear functionals
Best approximation of linear functionals

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This paper offers a Fortran program for the calculation of the recovery scheme of Micchelli, Rivlin and Winograd [15] which "recovers" a function from its values at certain data points in an "optimal" way. A short derivation of that recovery scheme is given first, as well as a derivation of the related envelope construction of Gaffney and Powell [8]. The underlying computational problem: to construct an extension with prescribed norm of a linear functional on some finite dimensional linear subspace to all of $\mathbb{L}_1[a,b]$.

L sub 1 (a, b)

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED 221 200

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)